3.016 Problem Set #4.Group, 2010

- **G4-1**

In this problem, the displacement field near an edge dislocation will be visualized, and its elastic field will be calculated and visualized.

In[1]:=
```
Clear[x, y]
```

Define r:

In[2]:=
```
r = Sqrt[x^2 + y^2]
```

Out[2]=
$$\sqrt{x^2 + y^2}$$

In[3]:=
```
Clear[pois]
```

Note that pois is the Poisson ratio.
Now write the x-displacement field u(x,y):

In[4]:=
```
u  =  (b / (2 Pi)) (ArcTan[x, y]  + 1 / (2 (1 - pois)) x y / r^2)
```

Out[4]=
$$\frac{b\left(\frac{x\,y}{2\,(1-\text{pois})\,(x^2+y^2)} + \text{ArcTan}[x,\,y]\right)}{2\,\pi}$$

Normaliz u by b gives:

In[5]:=
```
uNorm = u / b
```

Out[5]=
$$\frac{\frac{x\,y}{2\,(1-\text{pois})\,(x^2+y^2)} + \text{ArcTan}[x,\,y]}{2\,\pi}$$

Now write the y-displacement field v(x,y):

In[6]:=
```
v  =  (b / (2 Pi))
   ((1 -  2 pois) / (2 (1 - pois)) Log[b / r]  +  1 / (2 (1 - pois)) y^2 / r^2)
```

Out[6]=
$$\frac{b\left(\frac{y^2}{2\,(1-\text{pois})\,(x^2+y^2)} + \frac{(1-2\,\text{pois})\,\text{Log}\left[\frac{b}{\sqrt{x^2+y^2}}\right]}{2\,(1-\text{pois})}\right)}{2\,\pi}$$

Normaliz v by b gives:

In[7]:=
```
vNorm = v / b
```

Out[7]=

$$\frac{\frac{y^2}{2\,(1-\text{pois})\,(x^2+y^2)} + \frac{(1-2\,\text{pois})\,\text{Log}\left[\frac{b}{\sqrt{x^2+y^2}}\right]}{2\,(1-\text{pois})}}{2\,\pi}$$

*If we let x → ξ*b, y → ψ*b, then we can have:*

In[8]:=
```
uNormalized = Simplify[uNorm /. {x → ξ * b, y → ψ * b}, Assumptions → b > 0]
```

Out[8]=

$$\frac{-\xi\,\psi + 2\,(-1+\text{pois})\,(\xi^2+\psi^2)\,\text{ArcTan}[\xi,\,\psi]}{4\,\pi\,(-1+\text{pois})\,(\xi^2+\psi^2)}$$

and

In[9]:=
```
vNormalized = Simplify[vNorm /. {x → ξ * b, y → ψ * b}, Assumptions → b > 0]
```

Out[9]=

$$-\frac{2\,\psi^2 + (-1+2\,\text{pois})\,(\xi^2+\psi^2)\,\text{Log}[\xi^2+\psi^2]}{8\,\pi\,(-1+\text{pois})\,(\xi^2+\psi^2)}$$

G4-2

Generate displacement field by an edge dislocation, assuming pois = 1/4 and b=1:

In[10]:=
```
exampleDisplacementField = {uNorm, vNorm} /. {pois → 1 / 4, b → 1}
```

Out[10]=

$$\left\{\frac{\frac{2\,x\,y}{3\,(x^2+y^2)} + \text{ArcTan}[x,\,y]}{2\,\pi},\ \frac{\frac{2\,y^2}{3\,(x^2+y^2)} + \frac{1}{3}\,\text{Log}\left[\frac{1}{\sqrt{x^2+y^2}}\right]}{2\,\pi}\right\}$$

In[11]:=
```
exampleDisplacementFieldN =
  {uNormalized, vNormalized} /. {pois → 1 / 4}
```

Out[11]=

$$\left\{-\frac{-\xi\,\psi - \frac{3}{2}\,(\xi^2+\psi^2)\,\text{ArcTan}[\xi,\,\psi]}{3\,\pi\,(\xi^2+\psi^2)},\ \frac{2\,\psi^2 - \frac{1}{2}\,(\xi^2+\psi^2)\,\text{Log}[\xi^2+\psi^2]}{6\,\pi\,(\xi^2+\psi^2)}\right\}$$

Plot the displacement field using vector plot:

In[12]:=

```
VectorPlot[exampleDisplacementField, {x, -6, 6}, {y, -6, 6}]
```

Out[12]=



In[13]:=
```
VectorPlot[exampleDisplacementFieldN, {ξ, -6, 6}, {ψ, -6, 6}]
```

Out[13]=



The region near zero is problematic because of the singularity.
We'll exclude that region from the plot.
Here is one way of doing it. This could also be done with PieceWise or with an If.

In[15]:=

```
VectorPlot[exampleDisplacementField, {x, -6, 6},
 {y, -6, 6}, RegionFunction → (Norm[{#1, #2}] ≥ 2 &),
 VectorScale → 0.75, PlotRange → {{-6, 6}, {-6, 6}}]
```

Out[15]=



In[16]:=
```
VectorPlot[exampleDisplacementFieldN, {ξ, -6, 6},
 {ψ, -6, 6}, RegionFunction → (Norm[{#1, #2}] ≥ 2 &),
 VectorScale → 0.75, PlotRange → {{-6, 6}, {-6, 6}}]
```

Out[16]=


The discontinuity from the extra lattice plane is clearly observable.

- **G4-3**

According to the given definition of the (small) strain components, compute them:

In[17]:=
```
dExx =
 Simplify[D[u, x], Assumptions → x ∈ Reals && y ∈ Reals && b ∈ Reals]
```

Out[17]=
$$\frac{b\,y\,\left((3-2\,\text{pois})\,x^2+(1-2\,\text{pois})\,y^2\right)}{4\,\pi\,(-1+\text{pois})\,\left(x^2+y^2\right)^2}$$

In[18]:=
```
dEyy =
 Simplify[D[v, y], Assumptions → x ∈ Reals && y ∈ Reals && b ∈ Reals]
```

Out[18]=
$$-\frac{b\,y\,\left((1+2\,\text{pois})\,x^2+(-1+2\,\text{pois})\,y^2\right)}{4\,\pi\,(-1+\text{pois})\,\left(x^2+y^2\right)^2}$$

In[19]:=
```
dExy = Simplify[(D[u, y] + D[v, x]) / 2,
  Assumptions → x ∈ Reals && y ∈ Reals && b ∈ Reals]
```

Out[19]=

$$\frac{b\,x\,(-x^2+y^2)}{4\,\pi\,(-1+\text{pois})\,(x^2+y^2)^2}$$

Note that both x and y should be real numbers. Make sure we take those into the assumptions for simplification.

G4-4

Here it is the relationship between Young's modulus (Y) and shear modulus (G):

In[20]:=
```
Y = 2 G ( 1 + pois)
```

Out[20]=

$$2\,G\,(1+\text{pois})$$

List the six equations that relate between stress and strain components:

In[21]:=
```
eq1 = exx == (sxx - pois (syy + szz)) / Y
```

Out[21]=

$$\text{exx} == \frac{\text{sxx}-\text{pois}\,(\text{syy}+\text{szz})}{2\,G\,(1+\text{pois})}$$

In[22]:=
```
eq2 = eyy == (syy - pois (sxx + szz)) / Y
```

Out[22]=

$$\text{eyy} == \frac{\text{syy}-\text{pois}\,(\text{sxx}+\text{szz})}{2\,G\,(1+\text{pois})}$$

In[23]:=
```
eq3 = ezz == (szz - pois (sxx + syy)) / Y
```

Out[23]=

$$\text{ezz} == \frac{-\text{pois}\,(\text{sxx}+\text{syy})+\text{szz}}{2\,G\,(1+\text{pois})}$$

In[24]:=
```
eq4 = eyz == (1 + pois) syz / Y
```

Out[24]=

$$\text{eyz} == \frac{\text{syz}}{2\,G}$$

In[25]:=
```
eq5 = ezx == (1 + pois) szx / Y
```

Out[25]=

$$\text{ezx} == \frac{\text{szx}}{2\,G}$$

In[26]:=
```
eq6 = exy == (1 + pois) sxy / Y
```

Out[26]=
$$exy == \frac{sxy}{2\,G}$$

Solve for stress components:

In[27]:=
```
sols = Solve[{eq1, eq2, eq3, eq4, eq5, eq6},
   {sxx, syy, szz, syz, szx, sxy}] // Simplify
```

Out[27]=
$$\left\{\left\{sxx \to \frac{2\,G\,(exx\,(-1+pois) - (eyy+ezz)\,pois)}{-1+2\,pois},\ syy \to -\frac{2\,G\,(eyy - eyy\,pois + (exx+ezz)\,pois)}{-1+2\,pois},\right.\right.$$
$$\left.\left. szz \to -\frac{2\,G\,(ezz + (exx+eyy)\,pois - ezz\,pois)}{-1+2\,pois},\ syz \to 2\,eyz\,G,\ szx \to 2\,ezx\,G,\ sxy \to 2\,exy\,G\right\}\right\}$$

Substitute strains as computed in G4-3:

In[28]:=
```
tmp = Simplify[sols /.
   {exx → dExx, eyy → dEyy, ezz → 0, ezx → 0, eyz → 0, exy → dExy}]
```

Out[28]=
$$\left\{\left\{sxx \to \frac{b\,G\,y\,(3\,x^2+y^2)}{2\,\pi\,(-1+pois)\,(x^2+y^2)^2},\ syy \to \frac{b\,G\,y\,(-x^2+y^2)}{2\,\pi\,(-1+pois)\,(x^2+y^2)^2},\right.\right.$$
$$\left.\left. szz \to \frac{b\,G\,pois\,y}{\pi\,(-1+pois)\,(x^2+y^2)},\ syz \to 0,\ szx \to 0,\ sxy \to \frac{b\,G\,x\,(-x^2+y^2)}{2\,\pi\,(-1+pois)\,(x^2+y^2)^2}\right\}\right\}$$

These are the expressions of stresses found in text books for an edge dislocation .

G4-5

Take a particular example for stresses:

In[29]:=
```
sigxy = (sxy /. tmp)[[1]] /. {b → 1, G → 1, pois → 1 / 4}
```

Out[29]=
$$-\frac{2\,x\,(-x^2+y^2)}{3\,\pi\,(x^2+y^2)^2}$$

In[30]:=
```
sigxx = (sxx /. tmp)[[1]] /. {b → 1, G → 1, pois → 1 / 4}
```

Out[30]=
$$-\frac{2\,y\,(3\,x^2+y^2)}{3\,\pi\,(x^2+y^2)^2}$$

In[31]:=
```
sigyy = (syy /. tmp)[[1]] /. {b → 1, G → 1, pois → 1/4}
```

Out[31]=

$$-\frac{2\,y\,\left(-x^2+y^2\right)}{3\,\pi\,\left(x^2+y^2\right)^2}$$

In[32]:=
```
sigzz = (szz /. tmp)[[1]] /. {b → 1, G → 1, pois → 1/4}
```

Out[32]=

$$-\frac{y}{3\,\pi\,\left(x^2+y^2\right)}$$

In[33]:=
```
hydroPressure = Simplify[-(sigxx + sigyy + sigzz)/3]
```

Out[33]=

$$\frac{5\,y}{9\,\pi\,x^2+9\,\pi\,y^2}$$

Now visualize stress field of sxx:

In[34]:=
```
ContourPlot[sigxx, {x, -15, 15}, {y, -15, 15},
 RegionFunction → (Norm[{#1, #2}] ≥ 2 &), ContourLabels → True]
```

Out[34]=



Now visualize stress field of sxy:

In[35]:=
```
ContourPlot[sigxy, {x, -30, 30}, {y, -30, 30},
 RegionFunction → (Norm[{#1, #2}] ≥ 2 &),
 ContourLabels → True, ColorFunction → "Rainbow"]
```

Out[35]=



Now visualize stress field of P:

In[36]:=
```
ContourPlot[hydroPressure, {x, -10, 10},
 {y, -10, 10}, RegionFunction → (Norm[{#1, #2}] ≥ 2 &),
 ContourLabels → True, ColorFunction → "DarkRainbow"]
```

Out[36]=